

rust-verification

pan

September 3, 2020

Contents

1	Definitions and Notations	2
1.1	Basic data types	2
1.2	Expressions	2
1.3	Commands	3
2	Semantics of Expressions	4
3	Semantics of Commands	7
4	Code examples	9
4.1	Invalidating a reference	10
4.2	Swap	11

theory *Definitions*

imports *Main HOL–Library.LaTeXsugar*

begin

We present an imperative programming language with unique references called Unique, aiming at modeling the semantics of mutable references of Rust. In Rust, (mutable) references borrow the ownership, the capacity to observe and modify the content they are referring to, for a certain amount of time. References can borrow not only from a local variable or a heap allocation but also from another reference. The latter case is called reborrowing. Reborrow forms a tree-like relation between references. While the borrow checker of the Rust compiler enforces the reborrow relation to be well-formed (mutable xor alias!) statically, Unique tracks it dynamically. Imagine running an abstract interpreter of Rust with a dynamic borrow checker. At every point of use of unique references, Unique asserts that the reference is in the reborrow relation (the reference is valid) and removes every other reference reborrowed from it so that it is the only valid unique reference to the location. We expect that this dynamic nature of Unique will help us deal with (type-) unsafe portion of Rust. Note that this project is greatly influenced by R. Jung’s Stacked Borrows[1].

1 Definitions and Notations

1.1 Basic data types

type-synonym *vname* = *string*
datatype *ty* = *Tbool* | *Tint* | *Tref ty*
type-synonym *tag* = *nat*
type-synonym *address* = *nat*
datatype *tag-kind* = *Unique*

References consists of the address of the referent and the tag. A unique tag is assigned to each reference on creation.

datatype *val* = *VBool bool* | *VInt int* | *Reference address tag*

1.2 Expressions

datatype *place* = *Var vname* | *Deref place*

Datatype *place* represents an access path to data. It corresponds to `Place` in https://doc.rust-lang.org/nightly/nightly-rustc/rustc_middle/mir/struct.Place.html

datatype *operand* = *Place place* | *Constant val*

Datatype *operand* describes a value inside an rvalue. *Place* denotes the current value at the place, while *Constant* denotes a constant value. *operand* corresponds to `Place` in https://doc.rust-lang.org/nightly/nightly-rustc/rustc_middle/mir/enum.Operand.html, though we simplified the differentiation between a move and a copy (i.e. `Unique` doesn't track the ownership).

datatype
 rvalue = *Use operand*
 | *Box operand*
 | *Ref place*
 | *Reborrow place*
 | *Plus operand operand*
 | *Less operand operand*
 | *Not operand*
 | *And operand operand*

Datatype *rvalue* corresponds to an expression in a usual programming language (as in the previous version of `Unique`). We chose the term *rvalue* because of parity with Rust MIR (https://doc.rust-lang.org/nightly/nightly-rustc/rustc_middle/mir/enum.Rvalue.html)

Note that *rvalue* is *not* recursive. Compound expressions need to be broken apart so that intermediate computation is stored to a variable.

The semantics of the constructs is as follows:

- $Box :: operand \Rightarrow rvalue$ allocates memory in the heap, initializing with the argument (`Box::new` in Rust). Box returns a new unique reference. Note that Box in MIR doesn't initialize the location but only allocates. The semantics will be adapted to MIR's one as we formalize uninitialized memory.
- $Ref :: place \Rightarrow rvalue$ creates a unique reference pointing to the place.
- $Reborrow :: place \Rightarrow rvalue$ creates a new unique reference reborrowing from the argument (`&mut *p` in Rust, although most of the reborrows are automatically inserted by Rust). (TODO: I guess `Use(Copy(mutable ref))` in MIR corresponds to `Reborrow` in Unique.)

1.3 Commands

datatype

```
com = Skip
| Assign place rvalue (- ::= - [1000, 61] 61)
| Seq com com          (-;;/ - [60, 61] 60)
| If rvalue com com     ((IF -/ THEN -/ ELSE -) [0, 0, 61] 61)
| WHILE rvalue com      ((WHILE -/ DO -) [0, 61] 61)
```

type-synonym $gamma = vname \Rightarrow address * tag$

Unique implicitly performs boxing to variables. In other words, every variable behaves as a root reference to a memory cell.

type-synonym $tags = tag\text{-}kind\ list$

type-synonym $borrow\text{-}list = tag\ list$

type-synonym $heap = (val * borrow\text{-}list)\ list$

fun $kill :: tag \Rightarrow borrow\text{-}list \Rightarrow borrow\text{-}list$ **where**

```
kill t [] = [] |
kill t (x # xs) = (if t = x then [x] else
  case kill t xs of
    [] => [] |
    xs => x # xs)
```

The function $kill$ calculates references to be invalidated by the use of the given reference. The following two lemmas show that the used reference will be the “leaf” of the borrow tree.

lemma $[simp]: t \notin set\ ts \implies kill\ t\ ts = []$

$\langle proof \rangle \langle proof \rangle \langle proof \rangle \langle proof \rangle \langle proof \rangle$

lemma $t \in set\ ts \implies \exists ts'. kill\ t\ ts = ts' @ [t]$

$\langle proof \rangle$

fun $kill\text{-}heap :: (address * tag) \Rightarrow heap \Rightarrow heap$ **where**

```
kill-heap (a, t) H = (let (v, ts) = H ! a in H[a := (v, kill t ts)])
```

abbreviation $kill\text{-}all :: (address * tag) list \Rightarrow heap \Rightarrow heap$ **where**
 $kill\text{-}all\ refs\ H == foldr\ kill\text{-}heap\ refs\ H$

fun $writable :: heap \Rightarrow (address * tag) \Rightarrow bool$ **where**
 $writable\ H\ (p, t) \longleftrightarrow p < length\ H \wedge t \in set\ (snd\ (H\ !\ p))$

definition $readable$ **where** $readable = writable$

Functions $writable :: (val \times nat\ list) list \Rightarrow nat \times nat \Rightarrow bool$ and $readable :: (val \times nat\ list) list \Rightarrow nat \times nat \Rightarrow bool$ determine whether the given reference can be used to write to/read from it. The validity of references is determined by the existence in the current borrow tree. In Unique, we allow only unique references. Thus $readable$ is equivalent to $writable$.

definition $allocated :: heap \Rightarrow address \Rightarrow bool$ **where**
 $allocated\ H\ a \longleftrightarrow a < length\ H$
 $\langle proof \rangle \langle proof \rangle \langle proof \rangle \langle proof \rangle \langle proof \rangle \langle proof \rangle \langle proof \rangle \langle proof \rangle$
end
theory Exp
imports $Definitions$
begin

2 Semantics of Expressions

This section presents the big-step semantics of expressions.

Introducing a locale here looks a good idea as it helps us formulate axioms among the heap access modifier. Expected axioms:

- Modifier must be idempotent: $m\ p\ (m\ p\ H) = m\ p\ H$.
- Modifier must not allocate/deallocate a memory cell: $|m\ p\ H| = |H|$.
- The content is left intact: $fst\ (m\ p\ H)_{[a]} = fst\ H_{[a]}$

type-synonym $modifier = (address * tag) \Rightarrow heap \Rightarrow heap\ option$

fun $write\text{-}access :: modifier$ **where**

$write\text{-}access\ (a, t)\ H =$
 $(if\ writable\ H\ (a, t)\ then$
 $\ let\ (v, ts) = H\ !\ a\ in$
 $\ Some\ (H[a := (v, kill\ t\ ts)])$
 $else\ None)$

fun $read\text{-}access :: modifier$ **where**

$read\text{-}access\ (a, t)\ H =$
 $(if\ readable\ H\ (a, t)\ then$
 $\ let\ (v, ts) = H\ !\ a\ in$
 $\ Some\ (H[a := (v, kill\ t\ ts)])$
 $else\ None)$

inductive *place-sem* ::

$\text{gamma} \Rightarrow \text{modifier} \Rightarrow \text{heap} * \text{place} \Rightarrow \text{heap} * \text{val} \Rightarrow \text{bool}$

$((-; - \vdash - \Downarrow_p -))$ **for** Γ m

where

Var: $\Gamma \vdash x = (a, t) \implies \Gamma; m \vdash (H, \text{Var } x) \Downarrow_p (H, \text{Reference } a \ t) \mid$

Deref: $\llbracket \Gamma; m \vdash (H, p) \Downarrow_p (H', \text{Reference } a \ t); m \ (a, t) \ H' = \text{Some } H'' \rrbracket \implies \Gamma; m \vdash (H, \text{Deref } p) \Downarrow_p (H'', \text{fst } (H'' ! a))$

The relation *place-sem* describes the semantics of places with respect to a heap access modifier m . $\Gamma; m \vdash (H, p) \Downarrow_p (H', v)$ means that a place expression p evaluates to a value v under the environment Γ and H , while the access has changed the state of the heap to H' .

We give a shorthand for read and write access to the place.

abbreviation *read-place-sem* :: $\text{gamma} \Rightarrow \text{heap} * \text{place} \Rightarrow \text{heap} * \text{val} \Rightarrow \text{bool}$

$((- \vdash_r - \Downarrow_p -))$

where

$\Gamma \vdash_r p \Downarrow_p v == \Gamma; \text{read-access} \vdash p \Downarrow_p v$

abbreviation *write-place-sem* :: $\text{gamma} \Rightarrow \text{heap} * \text{place} \Rightarrow \text{heap} * \text{val} \Rightarrow \text{bool}$

$((- \vdash_w - \Downarrow_p -))$

where

$\Gamma \vdash_w p \Downarrow_p v == \Gamma; \text{write-access} \vdash p \Downarrow_p v$

Note that every operand is considered as a read access to the place.

inductive *operand-sem* ::

$\text{gamma} \Rightarrow \text{heap} * \text{operand} \Rightarrow \text{heap} * \text{val} \Rightarrow \text{bool}$

$((- \vdash - \Downarrow_{op} -))$ **for** Γ

where

Place: $\Gamma \vdash_r (H, p) \Downarrow_p v \implies \Gamma \vdash (H, \text{Place } p) \Downarrow_{op} v \mid$

Constant: $\Gamma \vdash (H, \text{Constant } v) \Downarrow_{op} (H, v)$

inductive *rvalue-sem* ::

$\text{gamma} \Rightarrow \text{tags} * \text{heap} * \text{rvalue} \Rightarrow \text{tags} * \text{heap} * \text{val} \Rightarrow \text{bool}$

$((- \vdash - \Downarrow -))$ **for** Γ

where

Use: $\Gamma \vdash (H, op) \Downarrow_{op} (H', v) \implies \Gamma \vdash (T, H, \text{Use } op) \Downarrow (T, H', v) \mid$

Box: $\llbracket \Gamma \vdash (H, op) \Downarrow_{op} (H', v); p = \text{length } H'; t = \text{length } T \rrbracket$

$\implies \Gamma \vdash (T, H, \text{Box } op) \Downarrow (T @ [\text{Unique}], H' @ [(v, [t])], \text{Reference } p \ t) \mid$

Ref: $\Gamma \vdash_w (H, p) \Downarrow_p (H', \text{Reference } a \ t) \implies \Gamma \vdash (T, H, \text{Ref } p) \Downarrow (T, H', \text{Reference } a \ t) \mid$

Reborrow: $\llbracket \Gamma \vdash_w (H, p) \Downarrow_p (H', \text{Reference } a \ t); \text{writable } H \ (a, t);$

$t' = \text{length } T; H' ! a = (v, ts) \rrbracket$

$\implies \Gamma \vdash (T, H, \text{Reborrow } p) \Downarrow (T @ [\text{Unique}], H'[a := (v, ts @ [t'])],$

$\text{Reference } a \ t') \mid$

Plus: $\llbracket \Gamma \vdash (H, lhs) \Downarrow_{op} (H', \text{VInt } lhs'); \Gamma \vdash (H', rhs) \Downarrow_{op} (H'', \text{VInt } rhs') \rrbracket$

$\implies \Gamma \vdash (T, H, \text{Plus } lhs \ rhs) \Downarrow (T, H'', \text{VInt } (lhs' + rhs')) \mid$

Less: $\llbracket \Gamma \vdash (H, lhs) \Downarrow_{op} (H', \text{VInt } lhs'); \Gamma \vdash (H', rhs) \Downarrow_{op} (H'', \text{VInt } rhs') \rrbracket$

$\implies \Gamma \vdash (T, H, \text{Less } lhs \ rhs) \Downarrow (T, H'', \text{VBool } (lhs' < rhs')) \mid$

$$\begin{aligned}
& \text{Not: } \Gamma \vdash (H, op) \Downarrow_{op} (H', VBool\ v) \implies \Gamma \vdash (T, H, Not\ op) \Downarrow (T, H', VBool\ (\neg v)) \mid \\
& \text{And: } \llbracket \Gamma \vdash (H, lhs) \Downarrow_{op} (H', VBool\ lhs'); \Gamma \vdash (H', rhs) \Downarrow_{op} (H'', VBool\ rhs') \rrbracket \\
& \implies \Gamma \vdash (T, H, And\ lhs\ rhs) \Downarrow (T, H'', VBool\ (lhs' \wedge rhs'))
\end{aligned}$$

Some TODO notes on the semantics of expressions:

- I don't get the semantics of *Ref* in MIR and how it should behave under the auto-boxing of variables.
- I'm not sure readable/writable checking is correct.

We present the intuition behind the selected rules. First, *Box* rule allocates a new slot at the end of heap, returning a reference pointing to the slot with a fresh tag.

$$\frac{\Gamma \vdash (H, op) \Downarrow_{op} (H', v) \quad p = |H'| \quad t = |T|}{\Gamma \vdash (T, H, Box\ op) \Downarrow (T @ [Unique], H' @ [(v, [t])], Reference\ p\ t)} \text{Box}$$

Next, *USE* and *REF* rules allow us to read the content through a place (or give a constant).

$$\frac{\Gamma \vdash (H, op) \Downarrow_{op} (H', v)}{\Gamma \vdash (T, H, Use\ op) \Downarrow (T, H', v)} \text{USE}$$

$$\frac{\Gamma \vdash_w (H, p) \Downarrow_p (H', Reference\ a\ t)}{\Gamma \vdash (T, H, Ref\ p) \Downarrow (T, H', Reference\ a\ t)} \text{REF}$$

Last but not least, *REBORROW* rule creates a new reference pointing to the same object but with a fresh tag. The newly created reference is writable. Thus the original reference must be valid for writes. Reborrow is considered as the use of the original reference and hence it invalidates former children of the reference. As a result, the reborrow tree will have the original reference and the reborrowing reference as the last two element.

$$\begin{array}{c}
\text{REBORROW RULE:} \\
\Gamma \vdash_w (H, p) \Downarrow_p (H', Reference\ a\ t) \\
writable\ H\ (a, t) \quad t' = |T| \quad H'_{[a]} = (v, ts)
\end{array}
\frac{}{\Gamma \vdash (T, H, Reborrow\ p) \Downarrow (T @ [Unique], H'[a := (v, ts @ [t'])], Reference\ a\ t')}$$

We can prove that the semantics of expressions is deterministic.

lemma *place-sem-det*: $\Gamma; m \vdash p \Downarrow_p v \implies \Gamma; m \vdash p \Downarrow_p v' \implies v' = v$
 $\langle \text{proof} \rangle$

lemma *place-sem-det'*: $\Gamma; m \vdash (H, p) \Downarrow_p (H', v') \implies \Gamma; m \vdash (H, p) \Downarrow_p (H'', v'') \implies (H'', v'') = (H', v')$
 $\langle \text{proof} \rangle$

lemma *operand-sem-det*: $\Gamma \vdash op \Downarrow_{op} v \implies \Gamma \vdash op \Downarrow_{op} v' \implies v' = v$
 $\langle \text{proof} \rangle$

lemma *operand-sem-det'*: $\Gamma \vdash (H, op) \Downarrow_{op} (H', v') \implies \Gamma \vdash (H, op) \Downarrow_{op} (H'', v'') \implies (H'', v'') = (H', v')$
 $\langle \text{proof} \rangle$

lemma *rvalue-sem-det*: $\Gamma \vdash (T, H, rv) \Downarrow (T', H', v') \implies \Gamma \vdash (T, H, rv) \Downarrow (T'', H'', v'') \implies (T'', H'', v'') = (T', H', v')$
 $\langle \text{proof} \rangle$

The following lemmas are sanity checks of the semantics. We must be able to write through the references retrieved by BOX and REBORROW rules.

lemma *box-writable*: $\Gamma \vdash (T, H, \text{Box } e) \Downarrow (T', H', \text{Reference } a \ t) \implies \text{writable } H' \ (a, t)$
 $\langle \text{proof} \rangle$

lemma *wa-preserve-length*: $\text{write-access at } H = \text{Some } H' \implies \text{length } H' = \text{length } H$
 $\langle \text{proof} \rangle$

lemma *write-preserve-length*: $\Gamma \vdash_w p \Downarrow_p v \implies \text{length } (\text{fst } p) = \text{length } (\text{fst } v)$
 $\langle \text{proof} \rangle$

lemma *reborrow-writable*: $\Gamma \vdash (T, H, \text{Reborrow } p) \Downarrow (T', H', \text{Reference } a \ t) \implies \text{writable } H' \ (a, t)$
 $\langle \text{proof} \rangle$

end

$\langle \text{proof} \rangle \langle \text{proof} \rangle$

theory *Unique*

imports *Main Definitions Exp Star*

begin

3 Semantics of Commands

This section describes the small-step semantics of commands. The configuration of an execution consists of:

- the tags issued T (of type *tags*);

- the heap H (of type *heap*) which holds the actual data and the reborrow trees for each address; and
- the command to be executed c (of type *com*). The variable environment Γ (of type *gamma*) is fixed throughout the execution.

type-synonym $config = tags * heap * com$

inductive $com-sem :: gamma \Rightarrow config \Rightarrow config \Rightarrow bool ((- \vdash - \rightarrow -))$ **for** Γ

where

Assign: $\llbracket \Gamma \vdash_w (H, p) \Downarrow_p (H', \text{Reference } a \ t); \text{writable } H' (a, t); H'' = \text{kill-heap } (a, t) \ H' \rrbracket$

$\Gamma \vdash (T, H'', rv) \Downarrow (T', H''', v); H''' ! a = (-, ts) \rrbracket$

$\implies \Gamma \vdash (T, H, p ::= rv) \rightarrow (T', H''[a := (v, ts)], \text{Skip}) \mid$

SeqL: $\Gamma \vdash (T, H, \text{Skip};; c) \rightarrow (T, H, c) \mid$

SeqR: $\Gamma \vdash (T, H, c1) \rightarrow (T', H', c') \implies \Gamma \vdash (T, H, c1;; c2) \rightarrow (T', H', c';; c2) \mid$

IfTrue: $\Gamma \vdash (T, H, b) \Downarrow (T', H', \text{VBool True})$

$\implies \Gamma \vdash (T, H, \text{IF } b \text{ THEN } c1 \text{ ELSE } c2) \rightarrow (T', H', c1) \mid$

IfFalse: $\Gamma \vdash (T, H, b) \Downarrow (T', H', \text{VBool False})$

$\implies \Gamma \vdash (T, H, \text{IF } b \text{ THEN } c1 \text{ ELSE } c2) \rightarrow (T', H', c2) \mid$

While: $\Gamma \vdash (T, H, \text{WHILE } b \text{ DO } c) \rightarrow (T, H, \text{IF } b \text{ THEN } (c;; \text{WHILE } b \text{ DO } c) \text{ ELSE } \text{Skip})$

abbreviation $com-sem-steps :: gamma \Rightarrow config \Rightarrow config \Rightarrow bool ((- \vdash - \rightarrow^* -))$

where

$\Gamma \vdash cfg \rightarrow^* cfg' == \text{star } (com-sem \ \Gamma) \ cfg \ cfg'$

Figure 1 shows the assignment rules. Other commands are standard.

The ASSIGN rule assigns a value to the the location. First, $\Gamma \vdash_w (H, p) \Downarrow_p (H', \text{Reference } a \ t)$ computes the location with write modifier. Next, $\text{writable } H' (a, t)$ and $H'' = \text{kill-heap } (a, t) \ H'$ assert that the reference is writable and unique. Note that this assertion is made *before* the evaluation of the right hand side. This means Unique does not support Two Phase Borrows currently. Finally, we compute the right hand side and assigns the result to the location.

$$\frac{\begin{array}{l} \Gamma \vdash_w (H, p) \Downarrow_p (H', \text{Reference } a \ t) \\ \text{writable } H' (a, t) \quad H'' = \text{kill-heap } (a, t) \ H' \\ \Gamma \vdash (T, H'', rv) \Downarrow (T', H''', v) \quad H'''[a] = (uu, ts) \end{array}}{\Gamma \vdash (T, H, p ::= rv) \rightarrow (T', H''[a := (v, ts)], \text{Skip})} \text{ASSIGN}$$

Figure 1: The assignment rule

The following lemmas show that the execution of the commands is deterministic.

definition *final* :: *config* \Rightarrow *bool* **where**
final cfg \longleftrightarrow (*case cfg of* (\neg , \neg , *Skip*) \Rightarrow *True* | \neg \Rightarrow *False*)

definition *stuck* :: *gamma* \Rightarrow *config* \Rightarrow *bool* **where**
stuck Γ *cfg* \longleftrightarrow (*case cfg of*
 $(\neg, \neg, \text{Skip}) \Rightarrow \text{False}$ |
 $\neg \Rightarrow \neg(\exists \text{cfg}'. (\Gamma \vdash \text{cfg} \rightarrow \text{cfg}'))$)

lemma *skip-final-noteq[simp]*: $\Gamma \vdash (T, H, c) \rightarrow (T, H', c') \Longrightarrow c \neq \text{Skip}$
 $\langle \text{proof} \rangle$

lemma *skip-final[simp]*: $\Gamma \vdash (T, H, \text{Skip}) \rightarrow (T', H', c) \Longrightarrow \text{False}$
 $\langle \text{proof} \rangle$

lemma *com-sem-det*: $\Gamma \vdash \text{cfg} \rightarrow \text{cfg}' \Longrightarrow \Gamma \vdash \text{cfg} \rightarrow \text{cfg}'' \Longrightarrow \text{cfg}'' = \text{cfg}'$
 $\langle \text{proof} \rangle$

We show the transitivity of *com-sem-steps*.

lemma *com-seql-trans*:
 $\Gamma \vdash (T, H, c1) \rightarrow^* (T', H', c1') \Longrightarrow \Gamma \vdash (T, H, c1;; c2) \rightarrow^* (T', H', c1';; c2)$
 $\langle \text{proof} \rangle$

4 Code examples

We need to assign a pre-allocated location to each variable to run a program. Currently we initialize those locations with zeros, but they will be uninitialized when we implement uninitialized slots.

abbreviation $\Gamma_0 == \lambda\cdot. \text{undefined}$
fun *preallocate'* :: *vname list* \Rightarrow *gamma* * *tags* * *heap* \Rightarrow *gamma* * *tags* * *heap* **where**
preallocate' [] *ret* = *ret* |
preallocate' (*x* # *xs*) (Γ , *T*, *H*) =
 (*let* *t* = *length T* *in*
 let $\Gamma' = \Gamma(x := (t, t))$ *in*
 let *T'* = *T* @ [*Unique*] *in*
 let *H'* = *H* @ [(*VInt* 0, [*t*])] *in*
 preallocate' *xs* (Γ' , *T'*, *H'*))
fun *preallocate* :: *vname list* \Rightarrow *gamma* * *tags* * *heap* **where**
preallocate *xs* = *preallocate'* *xs* (Γ_0 , [], [])

value *preallocate* ["x", "y"]
value *map* (*fst* (*preallocate* ["x", "y"])) ["x", "y"]

Let Isabelle run Unique programs by generating code for the semantics.

code-pred [*show-modes*] *place-sem* $\langle \text{proof} \rangle$
code-pred [*show-modes*] *operand-sem* $\langle \text{proof} \rangle$

code-pred $[show-modes]$ *rvalue-sem* $\langle proof \rangle$
code-pred $[show-modes]$ *com-sem* $\langle proof \rangle$

4.1 Invalidating a reference

Consider the following Rust program.

```

1  let mut root = Box::new(42);
2  let mut px = &mut *root; // reborrowing from root
3  *px = 100;
4  let mut py = &mut *root; // another reborrow. invalidates px
5  *py = 200;
6  // *px = 300 // this write is invalid.

```

In line 1, we create a box that contains 42 in the heap. In the following two lines, we reborrow from it and use the reference to write. The interesting part is line 4. In this line, we create a new reference from `root`. As this reborrow asserts that `py` is the unique reference to the box, `px` must be invalidated at this point. Therefore, if we remove the comment in line 6, the program will perform an invalid write (thus `rustc` will reject the program). Let's execute the program with `Unique`. The following command is the translation of the program above without the last commented line.

definition $XY :: com$ **where**

$$\begin{aligned}
 XY = & \\
 & (Var \text{ "root" } ::= (Use \circ Constant \circ VInt) \ 42;; \\
 & (Var \text{ "px" } ::= (Reborrow \circ Var) \text{ "root" };; \\
 & ((Deref \circ Var) \text{ "px" } ::= (Use \circ Constant \circ VInt) \ 100;; \\
 & (Var \text{ "py" } ::= (Reborrow \circ Var) \text{ "root" };; \\
 & ((Deref \circ Var) \text{ "py" } ::= (Use \circ Constant \circ VInt) \ 200
 \end{aligned}$$

We can let Isabelle compute the program.

value *preallocate* $[\text{"root"}, \text{"px"}, \text{"py"}]$
abbreviation $\Gamma == \Gamma_0(\text{"root"} := (0, 0), \text{"px"} := (1, 1), \text{"py"} := (2, 2))$
abbreviation $T == [Unique, Unique, Unique]$
abbreviation $H == [(VInt \ 0, [0]), (VInt \ 0, [1]), (VInt \ 0, [2])]$
lemma *preallocate* $[\text{"root"}, \text{"px"}, \text{"py"}] = (\Gamma, T, H) \ \langle proof \rangle$

values $\{(T', H') \mid$
 $T' \ H' \ c'. \Gamma \vdash (T, H, XY) \rightarrow^* (T', H', c')\}$

The following shows the tags and the heap at the end of execution.

abbreviation $T_{XY} :: tags$ **where**

$$T_{XY} == [Unique, Unique, Unique, Unique, Unique]$$

abbreviation $H_{XY} :: heap$ **where**

$$H_{XY} == [(VInt \ 200, [0, 4]), (Reference \ 0 \ 3, [1]), (Reference \ 0 \ 4, [2])]$$

We can also prove that the program result in the state shown above in theory, but it's really tedious (why can't Isabelle prove it by computing the execution?). The actual proof is left to the reader.

lemma *final-xy*: $\Gamma \vdash (T, H, XY) \rightarrow^* (T_{XY}, H_{XY}, Skip)$
sorry

We show that accessing through **px** after the program will stuck.

lemma *intermediate-xyx*: $\Gamma \vdash (T, H, XY;; ((Deref \circ Var) \text{"px"}) ::= (Use \circ Constant \circ VInt) 300)$
 $\rightarrow^* (T_{XY}, H_{XY}, ((Deref \circ Var) \text{"px"}) ::= (Use \circ Constant \circ VInt) 300)$
 $\langle proof \rangle$

lemma *stuck-xyx*: *stuck* Γ
 $(T_{XY}, H_{XY}, ((Deref \circ Var) \text{"px"}) ::= (Use \circ Constant \circ VInt) 300)$
 $\langle proof \rangle$

lemma $\exists \text{cfg}. (\Gamma \vdash (T, H, XY;; ((Deref \circ Var) \text{"px"}) ::= (Use \circ Constant \circ VInt) 300) \rightarrow^* \text{cfg})$
 $\wedge \text{stuck } \Gamma \text{ cfg}$
 $\langle proof \rangle$

4.2 Swap

The following program swaps the content the given two references pointing to.

fun *swap* :: *place* \Rightarrow *place* \Rightarrow *com* **where**
swap *x* *y* =
 (*Var* *"reborrow-x"*) ::= *Reborrow* *x*;;
 (*Var* *"reborrow-y"*) ::= *Reborrow* *y*;;
 (*Var* *"tmp"*) ::= (*Use* \circ *Place* \circ *Deref* \circ *Deref* \circ *Var*) *"reborrow-x"*;;
 ((*Deref* \circ *Var*) *"reborrow-x"*) ::= (*Use* \circ *Place* \circ *Deref* \circ *Deref* \circ *Var*)
"reborrow-y";;
 ((*Deref* \circ *Var*) *"reborrow-y"*) ::= (*Use* \circ *Place* \circ *Deref* \circ *Deref* \circ *Var*) *"tmp"*

We can see the content swapped during the execution by running it on Isabelle.

values $\{H \mid T \ H \ c.$
 $\Gamma_0(\text{"reborrow-x"} := (0, 0), \text{"reborrow-y"} := (1, 1), \text{"tmp"} := (2, 2), \text{"a"} :=$
 $(3, 3), \text{"b"} := (4, 4)) \vdash$
 $[[Unique, Unique, Unique],$
 $[(VInt\ 0, [0]), (VInt\ 0, [1]), (VInt\ 0, [2]), (VInt\ 10000, [3]), (VInt\ 20000, [4])],$
 $swap\ (Var\ \text{"a"})\ (Var\ \text{"b"})] \rightarrow^* (T, H, c)\}$

Proving the correctness of *swap* would need a Hoare Logic. The road continues...

end
theory *Rustv*

```

imports Simpl.Vcg Simpl.Simpl-Heap
begin

datatype rust-error = invalid-ref

type-synonym tag = nat
datatype val = int-val int
record tagged-ref =
  pointer :: ref
  tag :: tag

record globals-ram =
  memory :: val list
  tags :: tag list list
  issued-tags :: tag list

fun wf-tags :: tag list list  $\Rightarrow$  bool where
  wf-tags [] = True |
  wf-tags ([] # -) = False |
  wf-tags (- # t) = wf-tags t
lemma wf-tags-spec: wf-tags ts  $\longleftrightarrow$  ( $\forall t \in \text{set } ts. t \neq []$ )
   $\langle \text{proof} \rangle$ 

lemma [simp, intro]:  $\llbracket \text{wf-tags } ts; x \neq [] \rrbracket \Longrightarrow \text{wf-tags } (ts @ [x])$ 
   $\langle \text{proof} \rangle$ 

fun collect-tags :: tag list list  $\Rightarrow$  tag set where
  collect-tags ts = foldr ( $\lambda ts \text{ accum. set } ts \cup \text{accum}$ ) ts {}
lemma collect-tags-spec:  $t \in \text{collect-tags } ts \longleftrightarrow (\exists i < \text{length } ts. t \in \text{set } (ts ! i))$ 
   $\langle \text{proof} \rangle$ 

declare collect-tags.simps[simp del]

lemma collect-tags-update[simp, intro]:
   $t \in \text{collect-tags } (ts[p := x]) \Longrightarrow t \in \text{collect-tags } ts \vee t \in \text{set } x$ 
   $\langle \text{proof} \rangle$ 

lemma [simp, intro]:  $\text{collect-tags } (ts @ [t]) = \text{set } t \cup (\text{collect-tags } ts)$ 
   $\langle \text{proof} \rangle$ 

lemma [simp, intro]: finite (collect-tags ts)
   $\langle \text{proof} \rangle$ 

fun wf-heap :: 'a globals-ram-scheme  $\Rightarrow$  bool where
  wf-heap s  $\longleftrightarrow$ 
    length (memory s) = length (tags s)
     $\wedge$  wf-tags (tags s)
     $\wedge$  collect-tags (tags s)  $\subseteq \text{set } (\text{issued-tags } s)$ 

```

fun *invalidate-children* :: *tagged-ref* \Rightarrow 'a *globals-ram-scheme* \Rightarrow 'a *globals-ram-scheme*
where

invalidate-children *r s* =
 (let *p* = *Rep-ref* (*pointer* *r*);
 ts = *tags* *s* ! *p*;
 ts' = *dropWhile* ((\neq) (*tag* *r*)) *ts* in
s [*tags* := (*tags* *s*) [*p* := *ts'*]])

fun-cases *invalidate-childrenE*: *invalidate-children* *r s* = *s'*

lemma *dropWhile-hd-eq*[*simp*, *intro*]: $x \in \text{set } xs \implies \text{hd } (\text{dropWhile } ((\neq) \ x) \ xs) = x$
 <proof>

lemma *dropWhile-in*[*simp*, *intro*]: $x \in \text{set } xs \implies x \in \text{set } (\text{dropWhile } ((\neq) \ x) \ xs)$
 <proof>

lemma $\llbracket \text{invalidate-children } r \ s = s'; \text{Rep-ref } (\text{pointer } r) < \text{length } (\text{tags } s); \text{tag } r \in \text{set } ((\text{tags } s') ! \text{Rep-ref } (\text{pointer } r)) \rrbracket$
 $\implies \text{hd } ((\text{tags } s') ! \text{Rep-ref } (\text{pointer } r)) = \text{tag } r$
 <proof>

lemma *invalidate-children-memory*[*intro*]:
invalidate-children *r s* = *s'* $\implies \text{memory } s' = \text{memory } s$
 <proof>

fun *memwrite*
 :: *tagged-ref* \Rightarrow *val* \Rightarrow 'a *globals-ram-scheme* \Rightarrow 'a *globals-ram-scheme*
where

memwrite *p v s* =
 (let *memory* = *memory* *s* in
 let *memory'* = *memory* [*Rep-ref* (*pointer* *p*) := *v*] in
s [*memory* := *memory'*])

lemma *memwrite-written*[*simp*]:
fixes *p v s s'*
assumes $s' = \text{memwrite } p \ v \ s$
 $\text{Rep-ref } (\text{pointer } p) < \text{length } (\text{memory } s)$
shows $(\text{memory } s') ! \text{Rep-ref } (\text{pointer } p) = v$
 <proof>

lemma *memwrite-not-written*[*simp*]:
fixes *p p' v s s'*
assumes $s' = \text{memwrite } p \ v \ s$
 $\text{pointer } p \neq \text{pointer } p'$
shows $(\text{memory } s') ! \text{Rep-ref } (\text{pointer } p') = (\text{memory } s) ! \text{Rep-ref } (\text{pointer } p')$
 <proof>

lemma *memwrite-tags*:

```

fixes  $p\ v\ s\ s'$ 
assumes  $s' = \text{memwrite } p\ v\ s$ 
shows  $\text{tags } s' = \text{tags } s$ 
 $\langle \text{proof} \rangle$ 

fun  $\text{new-tag} :: 'a\ \text{globals-ram-scheme} \Rightarrow \text{tag}$  where
   $\text{new-tag } s = \text{fold } (\lambda t\ \text{accum}. \text{max } t\ \text{accum})\ (\text{issued-tags } s)\ 0 + 1$ 

fun-cases  $\text{new-tag-elim}$ :  $\text{new-tag } s = t$ 

lemma  $[\text{simp}, \text{intro}]$ :  $\text{new-tag } s = t \implies t \notin \text{set } (\text{issued-tags } s)$ 
 $\langle \text{proof} \rangle$ 

lemma  $\text{fold-max-init}[\text{intro}]$ :  $\text{fold max } xs\ (n :: \text{nat}) = m \implies m \geq n$ 
 $\langle \text{proof} \rangle$ 

lemma  $\text{fold-max-elem}[\text{intro}]$ :  $\text{fold max } xs\ (n :: \text{nat}) = m \implies \forall x \in \text{set } xs. m \geq x$ 
 $\langle \text{proof} \rangle$ 

lemma  $\text{max-fold-max}$ :  $\forall x \in \text{set } xs. m \geq x \implies \text{fold max } xs\ m = m$ 
 $\langle \text{proof} \rangle$ 

lemma
assumes
   $\text{wf-heap } s$ 
   $\text{new-tag } s = t$ 
   $r < \text{length } (\text{tags } s)$ 
shows  $t \notin \text{set } (\text{tags } s ! r)$ 
 $\langle \text{proof} \rangle$ 

fun  $\text{writable} :: \text{tagged-ref} \Rightarrow 'a\ \text{globals-ram-scheme} \Rightarrow \text{bool}$  where
   $\text{writable } r\ s =$ 
     $(\text{let } p = \text{Rep-ref } (\text{pointer } r)\ \text{in}$ 
       $\text{let } t = \text{tag } r\ \text{in}$ 
       $p < \text{length } (\text{memory } s) \wedge t \in \text{set } (\text{tags } s ! p))$ 

lemma  $\text{writable-update}[\text{simp}]$ :  $\text{writable } r\ (\text{memwrite } r'\ v\ s) = \text{writable } r\ s$ 
 $\langle \text{proof} \rangle$ 

lemma  $\text{writable-invalidated}[\text{intro}]$ :  $\text{writable } r\ s \implies \text{writable } r\ (\text{invalidate-children } r\ s)$ 
 $\langle \text{proof} \rangle$ 

fun  $\text{new-pointer} :: 'a\ \text{globals-ram-scheme} \Rightarrow \text{ref}$  where
   $\text{new-pointer } s = \text{Abs-ref } (\text{length } (\text{memory } s))$ 

fun  $\text{heap-new} :: \text{val} \Rightarrow 'a\ \text{globals-ram-scheme} \Rightarrow \text{tagged-ref} * 'a\ \text{globals-ram-scheme}$ 
where
   $\text{heap-new } v\ s =$ 

```

$(\text{let } p = \text{new-pointer } s;$
 $\quad t = \text{new-tag } s \text{ in}$
 $(\text{let } \text{pointer} = p, \text{tag} = t \text{ in}$
 $\quad s(\text{memory} := \text{memory } s @ [v], \text{tags} := \text{tags } s @ [[t]], \text{issued-tags} := t \#$
 $\text{issued-tags } s \text{)))}$

fun-cases *heap-new-elim*: $\text{heap-new } v \ s = (r, s')$

lemma *heap-new-writable*: $\llbracket \text{wf-heap } s; \text{heap-new } v \ s = (r', s') \rrbracket \implies \text{writable } r' \ s'$
 $\langle \text{proof} \rangle$

lemma *heap-new-wf-heap-update*: $\llbracket \text{wf-heap } s; \text{heap-new } v \ s = (r', s') \rrbracket \implies \text{wf-heap}$
 s'
 $\langle \text{proof} \rangle$

fun *reborrow* :: $\text{tagged-ref} \Rightarrow 'a \text{ globals-ram-scheme} \Rightarrow \text{tagged-ref} * 'a \text{ globals-ram-scheme}$
where

$\text{reborrow } r \ s =$
 $\quad (\text{let } p = \text{Rep-ref } (\text{pointer } r) \text{ in}$
 $\quad \text{let } t = \text{new-tag } s \text{ in}$
 $\quad \text{let } \text{tags} = (\text{tags } s)[p := t \# ((\text{tags } s) ! p)] \text{ in}$
 $\quad (\text{let } \text{pointer} = \text{pointer } r, \text{tag} = t \text{ in }, s(\text{tags} := \text{tags}, \text{issued-tags} := t \# \text{issued-tags}$
 $\quad s \text{)))}$

fun-cases *reborrow-elim*: $\text{reborrow } r \ s = (r', s')$

lemma *reborrow-pointer*: $\text{reborrow } r \ s = (r', s') \implies \text{pointer } r' = \text{pointer } r$
 $\langle \text{proof} \rangle$

lemma *reborrow-update-heap*: $\llbracket \text{wf-heap } s; \text{writable } r \ s; \text{reborrow } r \ s = (r', s') \rrbracket \implies$
 $\text{wf-heap } s'$
 $\langle \text{proof} \rangle$

lemma *reborrow-writable*: $\llbracket \text{wf-heap } s; \text{writable } r \ s; \text{reborrow } r \ s = (r', s') \rrbracket \implies$
 $\text{writable } r \ s'$
 $\langle \text{proof} \rangle$

end

References

- [1] R. Jung, H. Dang, J. Kang, and D. Dreyer. Stacked borrows: an aliasing model for rust. *Proc. ACM Program. Lang.*, 4(POPL):41:1–41:32, 2020.